

Issue XIX: Modal Homotopy Type System

М.Е. Сохацький ¹

¹ Національний технічний університет України
ім. Ігоря Сікорського
26 листопада 2021

Анотація

Here is presented a reincarnation of **cubicaltt** called **Anders**.

3mict

1	Introduction	3
2	Syntax	3
3	Semantics	6
3.1	Universe Hierarchies	6
3.2	Dependent Types	7
3.3	Path Equality	8
3.4	Strict Equality	9
3.5	Glue Types	11
3.6	de Rham Stack	13
3.7	Inductive Types	14
3.8	Higher Inductive Types	14
3.9	Simplicial Types	14
4	Properties	15
4.1	Soundness and Completeness	15
4.2	Canonicity, Normalization and Totality	15
4.3	Consistency and Decidability	15
4.4	Conservativity and Initiality	15
5	Conclusion	15

1 Introduction

Anders is a Modal HoTT proof assistant based on: classical MLTT-80 [6] with 0, 1, 2, W types; CCHM [11] in CHM [2] flavour as cubical type system with hcomp/transp operations; HTS [8] strict equality on pretypes; infinitesimal [1] modality primitives for differential geometry purposes. We tend not to touch general recursive higher inductive schemes, instead we will try to express as much HIT as possible through Suspensions, Truncations, Quotients primitives built into type checker core. Anders also aims to support simplicial types Simplex along with Hopf Fibrations built into core for sphere homotopy groups processing. This modification is called **Dan**. Full stack of Groupoid Infinity languages is given at AXIO/1¹ homepage.

The HTS language proposed by Voevodsky exposes two different presheaf models of type theory: the inner one is homotopy type system presheaf that models HoTT and the outer one is traditional Martin-Löf type system presheaf that models set theory with UIP. The motivation behind this doubling is to have an ability to express semisimplicial types. Theoretical work on merging inner and outer languages was continued in 2LTT [9].

Installation. While we are on our road to Lean-like tactic language, currently we are at the stage of regular cubical HTS type checker with CHM-style primitives. You may try it from Github sources: groupoid/anders² or install through OPAM package manager. Main commands are **check** (to check a program) and **repl** (to enter the proof shell).

```
$ opam install anders
```

Anders is fast, idiomatic and educational (think of optimized Mini-TT). We carefully draw the favourite Lean-compatible syntax to fit 200 LOC in Menhir. The CHM kernel is 1K LOC. Whole Anders compiles under 1 second and checks all the base library under 1/3 of a second [i5-12400]. Anders proof assistant as Homotopy Type System comes with its own Homotopy Library³.

2 Syntax

The syntax resembles original syntax of the reference CCHM type checker cubicaltt, is slightly compatible with Lean syntax and contains the full set of Cubical Agda [10] primitives (except generic higher inductive schemes).

Here is given the mathematical pseudo-code notation of the language expressions that come immediately after parsing. The core syntax definition of HTS language corresponds to the type defined in OCaml module:

Further Menhir BNF notation will be used to describe the top-level language E parser.

¹<https://axio.groupoid.space>

²<https://github.com/groupoid/anders/>

³<https://anders.groupoid.space/lib/>

```

cosmos := Uj | Vk
var := var name | hole
pi := Π name E E | λ name E E | E E
sigma := Σ name E E | (E, E) | E.1 | E.2
0 := 0 | ind0 E E E
1 := 1 | * | ind1 E E E
2 := 2 | 02 | 12 | ind2 E E E
W := W ident E E | sup E E | indW E E
id := Id E | ref E | idJ E
path := Path E | Ei | E @ E
I := I | 0 | 1 | E ∨ E | E ∧ E | ¬E
part := Partial E E | [ (E = I) → E, ... ]
sub := inc E | ouc E | E [ I ↦ E ]
kan := transp E E | hcomp E
glue := Glue E | glue E | unglue E E
Im := Im E | Inf E | Join E | indIm E E

E := cosmos | var | MLTT | CCHM | Im
CCHM := path | I | part | sub | kan | glue
MLTT := pi | sigma | id

```

Keywords. The words of a top-level language, file or repl, consist of keywords or identifiers. The keywords are following: **module**, **where**, **import**, **option**, **def**, **axiom**, **postulate**, **theorem**, (,), [,], <, >, /, .1, .2, **Π**, **Σ**, ,, **λ**, **V**, **∨**, **∧**, **¬**, **+**, **@**, **PathP**, **transp**, **hcomp**, **zero**, **one**, **Partial**, **inc**, **×**, **→**, **:**, **:=**, **↦**, **U**, **ouc**, **interval**, **inductive**, **Glue**, **glue**, **unglue**.

Identifiers. Identifiers support UTF-8. Identifiers couldn't start with **:**, **-**, **→**. Sample identifiers: **¬-of-∨**, **1→1**, **is-?**, **=**, **\$~]**!005x, **∞**, **x→Nat**.

Modules. Modules represent files with declarations. More accurate, BNF notation of module consists of imports, options and declarations.

menhir

```

start (Module.file) file
start (Module.command) repl
repl : COLON IDENT exp1 EOF | COLON IDENT EOF | exp0 EOF | EOF
file : MODULE IDENT WHERE line* EOF
path : IDENT
line : IMPORT path+ | OPTION IDENT IDENT | declarations

```

Imports. The import construction supports file folder structure (without file extensions) by using reserved symbol **/** for hierarchy walking.

Options. Each option holds bool value. Language supports following options: 1) girard (enables **U : U**); 2) pre-eval (normalization cache); 3) impredicative (infinite hierarchy with impredicativity rule); In Anders you can enable or disable language core types, adjust syntaxes or tune inner variables of the type checker.

Declarations. Language supports following top level declarations: 1) axiom (non-computable declaration that breaks normalization); 2) postulate (alternative or inverted axiom that can preserve consistency); 3) definition (almost any

explicit term or type in type theory); 4) lemma (helper in big game); 5) theorem (something valuable or complex enough).

```
axiom isProp (A : U) : U
def isSet (A : U) : U
:=  $\Pi$  (a b : A) (x y : Path A a b), Path (Path A a b) x y
```

Sample declarations. For example, signature `isProp (A : U)` of type `U` could be defined as normalization-blocking axiom without proof-term or by providing proof-term as definition.

In this example `(A : U)`, `(a b : A)` and `(x y : Path A a b)` are called telescopes. Each telescope consists of a series of lenses or empty. Each lense provides a set of variables of the same type. Telescope defines parameters of a declaration. Types in a telescope, type of a declaration and a proof-terms are a language expressions `expl`.

menhir

```
ident : IRREF | IDENT
lense : LPARENS ident+ COLON expl RPARENS
telescope : lense telescope
params : telescope | []
declarations :
| DEF IDENT params DEFEQ expl
| DEF IDENT params COLON expl DEFEQ expl
| AXIOM IDENT params COLON expl
```

Expressions. All atomic language expressions are grouped by four categories: `exp0` (pair constructions), `exp1` (non neutral constructions), `exp2` (path and pi applications), `exp3` (neutral constructions).

menhir

```
face : LPARENS IDENT IDENT IDENT RPARENS }
part : face+ ARROW expl }
exp0 : expl COMMA exp0 | expl }
exp1 : LSQ separated(COMMA, part) RSQ }
| LAM telescope COMMA expl | PI telescope COMMA expl
| SIGMA telescope COMMA expl | LSQ IRREF ARROW expl RSQ
| LT ident+ GT expl | exp2 ARROW expl
| exp2 PROD expl | exp2
```

The LR parsers demand to define `exp1` as expressions that cannot be used (without a parens enclosure) as a right part of left-associative application for both `Path` and `Pi` lambdas. Universe indicies U_j (inner fibrant), V_k (outer pretypes) and S (outer strict omega) are using unicode subscript letters that are already processed in lexer.

menhir

```
exp2 : exp2 exp3 | exp2 APPFORMULA exp3 | exp3 }
exp3 : LPARENS exp0 RPARENS LSQ exp0 MAP exp0 RSQ }
| HOLE | PRE | KAN | IDJ exp3
| exp3 FST | exp3 SND | NEGATE exp3 | INC exp3
| exp3 AND exp3 | exp3 OR exp3 | ID exp3 | REF exp3
| OUC exp3 | PATHP exp3 | PARTIAL exp3 | IDENT
| IDENT LSQ exp0 MAP exp0 RSQ } | HCOMP exp3
| LPARENS exp0 RPARENS } | TRANSP exp3 exp3
```

3 Semantics

The idea is to have a unified layered type checker, so you can disable/enable any MLTT-style inference, assign types to universes and enable/disable hierarchies. This will be done by providing linking API for pluggable presheaf modules. We selected 5 levels of type checker awareness from universes and pure type systems up to synthetic language of homotopy type theory. Each layer corresponds to its presheaves with separate configuration for universe hierarchies.

```
def lang : U
:= inductive {
  | UNI: cosmos → lang
  | PI: pure lang → lang
  | SIGMA: total lang → lang
  | ID: strict lang → lang
  | PATH: homotopy lang → lang
  | GLUE: glue lang → lang
  | INDUCTIVE: w012 lang → lang
}
```

We want to mention here with homage to its authors all categorical models of dependent type theory: Comprehension Categories (Grothendieck, Jacobs), LCCC (Seely), D-Categories and CwA (Cartmell), CwF (Dybjer), C-Systems (Voevodsky), Natural Models (Awodey). While we can build some transports between them, we leave this exercise for our mathematical components library. We will use here the Coquand's notation for Presheaf Type Theories in terms of restriction maps.

3.1 Universe Hierarchies

Language supports Agda-style hierarchy of universes: prop, fibrant (U), interval pretypes (V) and strict omega with explicit level manipulation. All universes are bounded with preorder

$$Fibrant_j \prec Pretypes_k \tag{1}$$

in which j, k are bounded with equation:

$$j < k. \tag{2}$$

Large elimination to upper universes is prohibited. This is extendable to Agda model:

```
def cosmos : U
:= inductive {
  | fibrant: N
  | pretypes: N
}
```

The **Anders** model contains only fibrant U_j and pretypes V_k universe hierarchies.

3.2 Dependent Types

Definition 1 (Type). A type is interpreted as a presheaf A , a family of sets A_I with restriction maps $u \mapsto u f, A_I \rightarrow A_J$ for $f : J \rightarrow I$. A dependent type B on A is interpreted by a presheaf on category of elements of A : the objects are pairs (I, u) with $u : A_I$ and morphisms $f : (J, v) \rightarrow (I, u)$ are maps $f : J \rightarrow I$ such that $v = u f$. A dependent type B is thus given by a family of sets $B(I, u)$ and restriction maps $B(I, u) \rightarrow B(J, u f)$.

We think of A as a type and B as a family of presheves $B(x)$ varying $x : A$. The operation $\Pi(x : A)B(x)$ generalizes the semantics of implication in a Kripke model.

Definition 2 (Pi). An element $w : [\Pi(x : A)B(x)](I)$ is a family of functions $w_f : \Pi(u : A(J))B(J, u)$ for $f : J \rightarrow I$ such that $(w_f u)g = w_f g(u g)$ when $u : A(J)$ and $g : K \rightarrow J$.

```
def pure (lang : U) : U
:= inductive { pi: name → nat → lang → lang → pure lang
              | lambda: name → nat → lang → lang
              | app: lang → lang
              }
```

Definition 3 (Sigma). The set $\Sigma(x : A)B(x)$ is the set of pairs (u, v) when $u : A(I), v : B(I, u)$ and restriction map $(u, v) f = (u f, v f)$.

```
def total (lang : U) : U
:= inductive { sigma: name → lang → total lang
              | pair: lang → lang
              | fst: lang
              | snd: lang
              }
```

The presheaf with only Pi and Sigma is called **MLTT-72** [4]. Its internalization in **Anders** is as follows:

```
def MLTT (A : U) : U1
:= Σ (Π-form : Π (B : A → U), U)
    (Π-ctor1 : Π (B : A → U), Pi A B → Pi A B)
    (Π-elim1 : Π (B : A → U), Pi A B → Pi A B)
    (Π-comp1 : Π (B : A → U) (a : A) (f : Pi A B),
      = (B a) (Π-elim1 B (Π-ctor1 B f) a) (f a))
    (Π-comp2 : Π (B : A → U) (a : A) (f : Pi A B),
      = (Pi A B) f (λ (x : A), f x))
    (Σ-form : Π (B : A → U), U)
    (Σ-ctor1 : Π (B : A → U) (a : A) (b : B a), Sigma A B)
    (Σ-elim1 : Π (B : A → U) (p : Sigma A B), A)
    (Σ-elim2 : Π (B : A → U) (p : Sigma A B), B (pr1 A B p))
    (Σ-comp1 : Π (B : A → U) (a : A) (b : B a),
      = A a (Σ-elim1 B (Σ-ctor1 B a b)))
    (Σ-comp2 : Π (B : A → U) (a : A) (b : B a),
      = (B a) b (Σ-elim2 B (a, b)))
    (Σ-comp3 : Π (B : A → U) (p : Sigma A B),
      = (Sigma A B) p (pr1 A B p, pr2 A B p)), Unit
```

3.3 Path Equality

The fundamental development of equality inside MLTT provers led us to the notion of ∞ -groupoid as spaces. In this way Path identity type appeared in the core of type checker along with De Morgan algebra on built-in interval type.

```
def CCHM (lang: U) : U
:= inductive { pretype (n: nat)
  | PathP (_: lang) | PLam (_: lang) | PApp (f a: lang)
  | I | 0 | 1 | And (a b: lang) | Or (a b: lang) | Neg (_: lang)
  | Transp (a b: lang) | HComp (a b c d: lang)
  | Partial (_: lang) | PartialP (a b: lang) | System (_: lang)
  | Sub (a b c: lang) | Inc (a b: lang) | Ouc (: lang)
  | Glue (: lang) | GlueElem (a b c: lang) | Unglue (_: lang)
}
```

Definition 4 (Cubical Presheaf \mathbb{I}). The identity types modeled with another presheaf, the presheaf on Lawvere category of distributive lattices (theory of De Morgan algebras) denoted with $\square - \mathbf{I} : \square^{op} \rightarrow \text{Set}$.

Definition 5 (Properties of \mathbf{I}). The presheaf \mathbf{I} : i) has to distinct global elements 0 and 1 (B_1); ii) $\mathbf{I}(I)$ has a decidable equality for each I (B_2); iii) \mathbf{I} is tiny so the path functor $X \mapsto X^{\mathbf{I}}$ has right adjoint (B_3).; iv) \mathbf{I} has meet and join (connections).

Interval Pretypes. While having pretypes universe V with interval and associated De Morgan algebra $(\wedge, \vee, -, 0, 1, \mathbf{I})$ is enough to perform DNF normalization and proving some basic statements about path, including: contractability of singletons, homotopy transport, congruence, functional extensionality; it is not enough for proving β rule for Path type or path composition.

Generalized Transport. Generalized transport transp addresses first problem of deriving the computational β rule for Path types:

```
theorem Path $\beta$  (A : U) (a : A) (C : D A) (d: C a a (refl A a))
: Equ (C a a (refl A a)) d (J A a C d a (refl A a))
:=  $\lambda$  (A : U)
  (a : A)
  (C :  $\Pi$  (x : A) (y : A), PathP ((\ $\backslash$ _ ) A) x y  $\rightarrow$  U),
  (d : C a a ((\ $\backslash$ _ ) a)),
  (j) transp ((\ $\backslash$ _ ) C a a ((\ $\backslash$ _ ) a)) -j d
```

Transport is defined on fibrant types (only) and type checker should cover all the cases Note that $\text{transp}^i (\text{Path}^j A v w) \varphi u_0$ case is relying on comp operation which depends on hcomp primitive. Here is given the first part of Simon Huber equations [3] for **transp**:

```
transp $^i$  N  $\varphi$  u $_0$  =u $_0$ 
transp $^i$  U  $\varphi$  A =A
transp $^i$  ( $\Pi$  (x : A), B)  $\varphi$  u $_0$  v =transp $^i$  B(x/w)  $\varphi$  (u $_0$  w(i/0))
transp $^i$  ( $\Sigma$  (x : A), B)  $\varphi$  u $_0$  = (transp $^i$  A  $\varphi$  (u $_0$ .1), transp $^i$  B(x/v)  $\varphi$  (u $_0$ .2))
transp $^i$  (Path $^j$  v w)  $\varphi$  u $_0$  = (j) comp $^i$  A [ $\phi$  u $_0$  j, (j=0)  $\mapsto$  v, (j=1)  $\mapsto$  w] (u $_0$  j)
transp $^i$  (Glue [ $\varphi$   $\mapsto$  (T,w)] A)  $\psi$  u $_0$  =glue [ $\phi$ (i/1)  $\mapsto$  t' $_1$ ] a' $_1$  : B(i/1)
```


Partial Elements. In order to explicitly define `hcomp` we need to specify `n`-cubes where some faces are missing. Partial primitives `isOne`, `l=1` and `UIP` on pretypes are derivable in `Anders` due to landing strict equality `Id` in `V` universe. The idea is that `(Partial A r)` is the type of cubes in `A` that are only defined when `IsOne r` holds. `(Partial A r)` is a special version of the function space `IsOne r → A` with a more extensional equality: two of its elements are considered judgmentally equal if they represent the same subcube of `A`. They are equal whenever they reduce to equal terms for all the possible assignment of variables that make `r` equal to `1`.

```
def Partial' (A : U) (i : I) := Partial A i
def isOne : I → V := Id I 1
def l=1 : isOne 1 := ref 1
def UIP (A : V) (a b : A) (p q : Id A a b) : Id (Id A a b) p q := ref p
```

Cubical Subtypes. For `(A : U) (i : I) (Partial A i)` we can define subtype `A [i ↦ u]`. A term of this type is a term of type `A` that is definitionally equal to `u` when `(IsOne i)` is satisfied. We have forth and back fusion rules `ouc (inc v) = v` and `inc (ouc v) = v`. Moreover, `ouc v` will reduce to `u l=1` when `i=1`.

```
def sub' (A : U) (i : I) (u : Partial A i) : V := A [i ↦ u]
def inc' (A : U) (i : I) (a : A) : A [i ↦ [(i = 1) → a]] := inc A i a
def ouc' (A : U) (i : I) (u : Partial A i) (a : A [i ↦ u]) : A := ouc a
```

Homogeneous Composition. `hcomp` is the answer to second problem: with `hcomp` and `transp` one can express path composition, groupoid, category of groupoids (groupoid interpretation and internalization in type theory). One of the main roles of homogeneous composition is to be a carrier in [higher] inductive type constructors for calculating of homotopy colimits and direct encoding of CW-complexes. Here is given the second part of Simon Huber equations [3] for **hcomp**:

```
hcompi N [ϕ ↦ 0] 0 = 0
hcompi N [ϕ ↦ S u] (S u0) = S (hcompi N [ϕ ↦ u] u0)
hcompi U [ϕ ↦ E] A = Glue [ϕ ↦ (E(i/1), equivi E(i/1-i))] A
hcompi (Π (x : A), B) [ϕ ↦ u] u0 v = hcompi B(x/v) [ϕ ↦ u v] (u0 v)
hcompi (Σ (x : A), B) [ϕ ↦ u] u0 = (v(i/1), compi B(x/v) [ϕ ↦ u.2] u0.2)
hcompi (Pathj A v w) [ϕ ↦ u] u0 = (j) hcompi A [ϕ ↦ u j], (j=0) ↦ v, (j=1) ↦ w (u0 j)
hcompi (Glue [ϕ ↦ (T,w)] A) [ψ ↦ u] u0 = glue [ϕ ↦ u(i/1)] (unglue u(i/1))
```

3.4 Strict Equality

To avoid conflicts with path equalities which live in fibrant universes strict equalities live in pretypes universes.

```
def strict (lang : U) : U
:= inductive { Id: name → lang
             | ref: lang → lang
             | idJ: lang → lang → lang
             }
```

We use strict equality in `HTS` for pretypes and partial elements which live in `V`. The presheaf configuration with `Pi`, `Sigma` and `Id` is called **MLTT-75** [5]. The

presheaf configuration with Π , Σ , Id and Path is called **HTS** (Homotopy Type System).

3.5 Glue Types

The main purpose of Glue types is to construct a cube where some faces have been replaced by equivalent types. This is analogous to how `hcomp` lets us replace some faces of a cube by composing it with other cubes, but for Glue types you can compose with equivalences instead of paths. This implies the univalence principle and it is what lets us transport along paths built out of equivalences.

```
def glue (lang : U) : U
:= inductive { Glue: lang → lang → lang
              | glue: lang → lang
              | unglue: lang → lang
              }
```

Basic Fibrational HoTT core by Pelayo, Warren, and Voevodsky (2012).

```
def fiber (A B : U) (f : A → B) (y : B) : U := Σ (x : A), Path B y (f x)
def isEquiv (A B : U) (f : A → B) : U := Π (y : B), isContr (fiber A B f y)
def equiv (A B : U) : U := Σ (f : A → B), isEquiv A B f
def contrSingl (A : U) (a b : A) (p : Path A a b)
: Path (Σ (x : A), Path A a x) (a, ⟨i⟩ a) (b, p) := ⟨i⟩ (p @ i, ⟨j⟩ p @ i ∨ j)
def idIsEquiv (A : U) : isEquiv A A (id A)
:= λ (a : A), ((a, ⟨i⟩ a), λ (z : fiber A A (id A) a), contrSingl A a z.1 z.2)
def idEquiv (A : U) : equiv A A := (id A, isContrSingl A)
```

The notion of Univalence was discovered by Vladimir Voevodsky as forth and back transport between fibrational equivalence as contractability of fibers and homotopical multi-dimensional heterogeneous path equality. The `Equiv → Path` type is called Univalence type, where univalence intro is obtained by Glue type and elim (`Path → Equiv`) is obtained by sigma transport from constant map.

```
def univ-formation (A B : U) := equiv A B → PathP (⟨i⟩ U) A B
def univ-intro (A B : U) : univ-formation A B := λ (e : equiv A B),
⟨i⟩ Glue B (∂ i) [(i = 0) → (A, e), (i = 1) → (B, idEquiv B)]
def univ-elim (A B : U) (p : PathP (⟨i⟩ U) A B)
: equiv A B := transp (⟨i⟩ equiv A (p @ i)) 0 (idEquiv A)
def univ-computation (A B : U) (p : PathP (⟨i⟩ U) A B)
: PathP (⟨i⟩ PathP (⟨i⟩ U) A B) (univ-intro A B (univ-elim A B p)) p
:= ⟨j i⟩ Glue B (j ∨ ∂ i)
[(i = 0) → (A, univ-elim A B p), (i = 1) → (B, idEquiv B),
(j = 1) → (p @ i, univ-elim (p @ i) B (⟨k⟩ p @ (i ∨ k)))]
```

Similar to Fibrational Equivalence the notion of Retract/Section based Isomorphism could be introduced as forth-back transport between isomorphism and path equality. This notion is somehow canonical to all cubical systems and is called Unimorphism here.

```
def iso-Form (A B : U) : U1 := iso A B → PathP (⟨i⟩ U) A B
def iso-Intro (A B : U) : iso-Form A B
:= λ (x : iso A B), isoPath A B x.f x.g x.s x.t
def iso-Elim (A B : U) : PathP (⟨i⟩ U) A B → iso A B
:= λ (p : PathP (⟨i⟩ U) A B),
(coerce A B p, coerce B A (⟨i⟩ p @ -i),
trans-1-trans A B p, λ (a : A), ⟨k⟩ trans-trans-1 A B p a @-k, ★)
```

Orton-Pitts basis for univalence computability (2017):

```

def ua (A B : U) (p : equiv A B) : PathP (⟨i⟩ U) A B := univ-intro A B p
def ua-β (A B : U) (e : equiv A B) : Path (A →B) (trans A B (ua A B e)) e.1
:= ⟨i⟩ λ (x : A), (idfun=idfun '' B @ -i)
  (idfun=idfun '' B @ -i) ((idfun=idfun ' B @ -i) (e.1 x)) )

```

3.6 de Rham Stack

Stack de Rham or Infinitesimal Shape Modality is a basic primitive for proving theorems from synthetic differential geometry. This type-theoretical framework was developed for the first time by Felix Cherubini under the guidance of Urs Schreiber. The Anders prover implements the computational semantics of the de Rham stack.

```

def  $\iota$  (A : U) (a : A) :  $\mathfrak{S}$  A :=  $\mathfrak{S}$ -unit a
def  $\mu$  (A : U) (a :  $\mathfrak{S}$  ( $\mathfrak{S}$  A)) :=  $\mathfrak{S}$ -join a
def is-coreduced (A : U) : U := isEquiv A ( $\mathfrak{S}$  A) ( $\iota$  A)
def  $\mathfrak{S}$ -coreduced (A : U) : is-coreduced ( $\mathfrak{S}$  A)
:= isoToEquiv ( $\mathfrak{S}$  A) ( $\mathfrak{S}$  ( $\mathfrak{S}$  A)) ( $\iota$  ( $\mathfrak{S}$  A)) ( $\mu$  A)
  ( $\lambda$  (x :  $\mathfrak{S}$  ( $\mathfrak{S}$  A)),  $\langle i \rangle$ x) ( $\lambda$  (y :  $\mathfrak{S}$  A),  $\langle i \rangle$ y)
def ind- $\mathfrak{S}\beta$  (A : U) (B :  $\mathfrak{S}$  A  $\rightarrow$  U) (f :  $\Pi$  (a : A),  $\mathfrak{S}$  (B ( $\iota$  A a))) (a : A)
  : Path ( $\mathfrak{S}$  (B ( $\iota$  A a))) (ind- $\mathfrak{S}$  A B f ( $\iota$  A a)) (f a) :=  $\langle i \rangle$  f a
def ind- $\mathfrak{S}$ -const (A B : U) (b :  $\mathfrak{S}$  B) (x :  $\mathfrak{S}$  A)
  : Path ( $\mathfrak{S}$  B) (ind- $\mathfrak{S}$  A ( $\lambda$  (i :  $\mathfrak{S}$  A), B) ( $\lambda$  (i : A), b) x) b :=  $\langle i \rangle$  b

```

Coreduced induction and its β -quation.

```

def  $\mathfrak{S}$ -ind (A : U) (B :  $\mathfrak{S}$  A  $\rightarrow$  U) (c :  $\Pi$  (a :  $\mathfrak{S}$  A),
  is-coreduced (B a)) (f :  $\Pi$  (a : A), B ( $\iota$  A a)) (a :  $\mathfrak{S}$  A) : B a
:= (c a (ind- $\mathfrak{S}$  A B ( $\lambda$  (x : A),  $\iota$  (B ( $\iota$  A x)) (f x)) a)).1.1
def  $\mathfrak{S}$ -ind $\beta$  (A : U) (B :  $\mathfrak{S}$  A  $\rightarrow$  U) (c :  $\Pi$  (a :  $\mathfrak{S}$  A),
  is-coreduced (B a)) (f :  $\Pi$  (a : A), B ( $\iota$  A a)) (a : A)
  : Path (B ( $\iota$  A a)) (f a) (( $\mathfrak{S}$ -ind A B c f) ( $\iota$  A a))
:=  $\langle i \rangle$  sec-equiv (B ( $\iota$  A a)) ( $\mathfrak{S}$  (B ( $\iota$  A a)))
  ( $\iota$  (B ( $\iota$  A a)), c ( $\iota$  A a)) (f a) @-i

```

Geometric Modal HoTT Framework: Infinitesimal Proximity, Formal Disk, Formal Disk Bundle, Differential.

```

def  $\sim$  (X : U) (a x' : X) : U := Path ( $\mathfrak{S}$  X) ( $\iota$  X a) ( $\iota$  X x')
def  $\mathbb{D}$  (X : U) (a : X) : U :=  $\Sigma$  (x' : X),  $\sim$  X a x'
def inf-prox-ap (X Y : U) (f : X  $\rightarrow$  Y) (x x' : X) (p :  $\sim$  X x x')
  :  $\sim$  Y (f x) (f x') :=  $\langle i \rangle$   $\mathfrak{S}$ -app X Y f (p @ i)
def  $T^\infty$  (A : U) : U :=  $\Sigma$  (a : A),  $\mathbb{D}$  A a
def inf-prox-ap (X Y : U) (f : X  $\rightarrow$  Y) (x x' : X) (p :  $\sim$  X x x')
  :  $\sim$  Y (f x) (f x') :=  $\langle i \rangle$   $\mathfrak{S}$ -app X Y f (p @ i)
def d (X Y : U) (f : X  $\rightarrow$  Y) (x : X) ( $\varepsilon$  :  $\mathbb{D}$  X x)
  :  $\mathbb{D}$  Y (f x) := (f  $\varepsilon$ .1, inf-prox-ap X Y f x  $\varepsilon$ .1  $\varepsilon$ .2)
def  $T^\infty$ -map (X Y : U) (f : X  $\rightarrow$  Y) ( $\tau$  :  $T^\infty$  X) :  $T^\infty$  Y := (f  $\tau$ .1, d X Y f  $\tau$ .1  $\tau$ .2)

```

3.7 Inductive Types

Anders currently don't support Lean-compatible generic inductive schemes definition. So instead of generic inductive schemes Anders supports well-founded trees (W-types). Basic data types like List, Nat, Fin, Vec are implemented as W-types in base library.

- W, 0, 1, 2 basis of MLTT-80 (Martin-Löf)
- General Schemes of Inductive Types (Paulin-Mohring)

3.8 Higher Inductive Types

As for higher inductive types Anders has Three-HIT foundation (Coequalizer, Path Coequalizer and Colimit) to express other HITs. Also there are other foundations to consider motivated by typical tasks in homotopy (type) theory:

- Coequalizer, Path Coequalizer and Colimit (van der Weide)
- Suspension, Truncation, Quotient (Groupoid Infinity)
- General Schemes of Higher Inductive Types (Cubical Agda)

3.9 Simplicial Types

Modification of Anders with Simplicial types and Hopf Fibrations built into the core of type checker is called **Dan** with following recursive syntax (having f as Simplecies and coh as Path-coherence functions):

```
simplex n [v0 .. vn] { f0, f1, ..., fn | coh i1 i2 ... in } : Simplex
```

and instantiation example:

```
def s∞ : Simplicial
:= Π (v e : Simplex),
    δ10 =v, δ11 =v, s0 ⟨v,
    δ20 =e ∘ e, s10 ⟨δ20
    ⊢ ∞ (v, e, δ20 | δ10 δ11, s0, δ20, s10)
```

4 Properties

Soundness and completeness link syntax to semantics. Canonicity, normalization, and totality ensure computational adequacy. Consistency and decidability guarantee logical and practical usability. Conservativity and initiality support extensibility and universality.

4.1 Soundness and Completeness

Soundness is proven via cubical sets [11, 12, 13].

4.2 Canonicity, Normalization and Totality

Canonicity and normalization hold constructively [14, 15].

4.3 Consistency and Decidability

Consistency follows from the model [16]. Decidability is achieved for type checking [13].

4.4 Conservativity and Initiality

Conservativity and initiality is discussed by Shulman [18, 17]. Initiality is implicit in the syntactic construction [12].

5 Conclusion

This paper presents Anders, a proof assistant that reimplements cubicaltt within a Modal Homotopy Type System framework, based on MLTT-80 and CCHM/CHM. It integrates HTS strict equality, infinitesimal modalities, and primitives like suspensions or quotients, with the extension adding simplicial types and Hopf fibrations. Anders offers an efficient, idiomatic system — compiling in under one second — using a syntax of Lean and semantics of cubicaltt and Cubical Agda. As a practical refinement of cubicaltt, Anders serves as an accessible tool for homotopy type theory, with potential for incremental enhancements like a tactic language.

Список використаних джерел

- [1] Felix Cherubini. *Cartan Geometry in Modal Homotopy Type Theory*. 2019. <https://arxiv.org/pdf/1806.05966.pdf>.
- [2] Thierry Coquand, Simon Huber, and Anders Mörtberg. *On Higher Inductive Types in Cubical Type Theory*. 2017. <https://staff.math.su.se/anders.mortberg/papers/cubicalhits.pdf>.
- [3] Simon Huber. *On Higher Inductive Types in Cubical Type Theory*. 2017. <http://www.cse.chalmers.se/~simonhu/misc/hcomp.pdf>.
- [4] Per Martin-Löf. *An Intuitionistic Theory of Types*. 1972.
- [5] Per Martin-Löf. *An Intuitionistic Theory of Types: Predicative Part*. 1975.
- [6] Per Martin-Löf. *Intuitionistic Type Theory*. 1980. <https://raw.githubusercontent.com/michaelt/martin-lof/master/pdfs/Bibliopolis-Book-retypeset-1984.pdf>.
- [7] Christine Paulin-Mohring. *Introduction to the Calculus of Inductive Constructions*. 2015. <https://hal.inria.fr/hal-01094195/document>.
- [8] Vladimir Voevodsky. *A simple type system with two identity types*. 2013. <https://www.math.ias.edu/vladimir/sites/math.ias.edu/vladimir/files/HTS.pdf>.
- [9] Danil Annenkov, Paolo Capriotti, Nicolai Kraus, and Christian Sattler. *Two-Level Type Theory and Applications*. 2019. <https://arxiv.org/pdf/1705.03307.pdf>.
- [10] Andrea Vezzosi, Anders Mörtberg, and Andreas Abel. *Cubical Agda: A Dependently Typed Programming Language with Univalence and Higher Inductive Types*. 2019. <https://staff.math.su.se/anders.mortberg/papers/cubicalagda.pdf>.
- [11] Cyril Cohen, Thierry Coquand, Simon Huber, and Anders Mörtberg. *Cubical Type Theory: A Constructive Interpretation of the Univalence Axiom*. 2018. <https://staff.math.su.se/anders.mortberg/papers/cubicalagda.pdf>.
- [12] Steve Awodey. *Type Theory and Homotopy*. In *Epistemology versus Ontology: Essays on the Philosophy and Foundations of Mathematics in Honour of Per Martin-Löf*, pages 183–201. Springer, 2012.
- [13] Thierry Coquand. *A Survey of Constructive Models of Univalence*. 2018. Preprint or lecture notes.
- [14] Simon Huber. *Canonicity for Cubical Type Theory*. *Journal of Automated Reasoning*, 61(1–4):173–205, 2017.

- [15] Thomas Streicher. *Semantics of Type Theory: Correctness, Completeness and Independence Results*. Birkhäuser, Basel, 1991.
- [16] Marc Bezem, Thierry Coquand, and Simon Huber. *A Model of Type Theory in Cubical Sets*. arXiv preprint arXiv:1406.1731, 2014.
- [17] Michael Shulman. *Univalence for Inverse Diagrams and Homotopy Canonicity*. *Mathematical Structures in Computer Science*, 25(5):1203–1277, 2015.
- [18] Martin Hofmann. *Syntax and Semantics of Dependent Types*. In *Semantics and Logics of Computation*, pages 79–130. Cambridge University Press, 1997.